This program looks for solutions

$(a_1, a_2, \ldots a_n)$ for which the sum $\sum_{i=1}^{n} \frac{1}{a_i} = s$ holds,

where s is a rational value supplied by the user.

$$X[a, n, s] =$$

$$\left\{ (a_1, a_2, \ldots a_n) \in Z^n \;\middle|\; \sum_{i=1}^{n} \frac{1}{a_i} = s,\ a \le a_1 \le a_2 \le \ldots \le a_n,\ s \in Q \right\}.$$

```
X[a_, 0, s_] = If[s == 0, {{}}, {}];
X[a_, n_, 0] = {};
X[a_, n_, s_] := Union @@ Table[
   Prepend[#, a1] & /@ X[a1, n - 1, s - 1 / a1],
   {a1, Max[⌈1 / s⌉, a], ⌊n / s⌋}
  ]
```

```
X[2, 6, 2]
```

```
{{2, 2, 2, 3, 7, 42}, {2, 2, 2, 3, 8, 24}, {2, 2, 2, 3, 9, 18}, {2, 2, 2, 3, 10, 15},
 {2, 2, 2, 3, 12, 12}, {2, 2, 2, 4, 5, 20}, {2, 2, 2, 4, 6, 12}, {2, 2, 2, 4, 8, 8},
 {2, 2, 2, 5, 5, 10}, {2, 2, 2, 6, 6, 6}, {2, 2, 3, 3, 4, 12}, {2, 2, 3, 3, 6, 6},
 {2, 2, 3, 4, 4, 6}, {2, 2, 4, 4, 4, 4}, {2, 3, 3, 3, 3, 6}, {2, 3, 3, 3, 4, 4}, {3, 3, 3, 3, 3, 3}}
```

Y[a, n, s] is a subset of X[a, n, s]
where divisibility applies. In this case,
the numbers are such that a_i divides a_n for all $1 \le i \le n-1$.

```
Y[a_, n_, s_] := Select[X[a, n, s], AllTrue[#[[-1]] / #[[1 ;; -2]], IntegerQ] &]
```

```
Y[2, 6, 2]
```

```
{{2, 2, 2, 3, 7, 42}, {2, 2, 2, 3, 8, 24}, {2, 2, 2, 3, 9, 18},
 {2, 2, 2, 3, 12, 12}, {2, 2, 2, 4, 5, 20}, {2, 2, 2, 4, 6, 12},
 {2, 2, 2, 4, 8, 8}, {2, 2, 2, 5, 5, 10}, {2, 2, 2, 6, 6, 6}, {2, 2, 3, 3, 4, 12},
 {2, 2, 3, 3, 6, 6}, {2, 2, 4, 4, 4, 4}, {2, 3, 3, 3, 3, 6}, {3, 3, 3, 3, 3, 3}}
```

```
Y[5, 8, 3] // MatrixForm
```

```
{}
```